
PROBLEM: 
A metal bottle filled by pressurized air explodes. The bottle is axisymmetric but a full 
3D model is used to simulate the fragmentation. Two simulations are performed. In 
the first one only the structure is modelled and the inner pressure is applied by an 
IMPE PIMP boundary condition by means of CL3d and CL3T elements. In the 
second simulation, the inner and outer air are modelled by means of FL38 and FL36 
fluid elements and the fluid-structure interaction is modelled by the FLSR directive. 
In both calculations the fragmentation process is treated by means of DEBR particles. 
 
MESH: 
The bottle is meshed by CUB8 and PR6 solid elements. The fluid, when present, is 
modelled by means of FL38 and FL36 fluid elements. 
 
GEOMETRIC COMPLEMENTS 
Debris elements (DEBR) are associated with the whole bottle. By using PLEV 0, just 
one spherical particle replaces each failing element. 
 
MATERIALS: 
The metallic bottle uses a linear elastoplastic material (VM23) with an associated 
failure criterion (FAIL PEPS). For the fluid, the FLUT material is used. In the 
calculation with fluid, absorbing boundary conditions are applied on the external 
surface of the fluid domain by means of an IMPE ABSI material 
 
BOUNDARY CONDITIONS: 
The bottle is free in 3D space. Fluid-structure interaction in the second simulation 
uses the FLSR model. 

LOADING: 
In the first simulation an internal pressure raising in time is applied. In the second 
simulation, the loading is produced by the high initial pressure of the internal air. 
 
CALCULATIONS: 
Calculations are performed until a physical time of 0.35 ms, when the bottle is 
completely failed. 
 
RESULTS: 
The bottle fragments as expected. Some macro fragments survive until the end of the 
simulation, together with a cloud of DEBR particles. 
 
POST-TREATMENT 
Animations are produced. 
 
Numerical Solutions 
 
BOTT02 
 
This is the first simulation, including only the structure. The Cast3m mesh generation 
file reads: 
 
opti echo 0; 
opti donn 'pxordpoi.proc'; 
opti donn 'pxvolu3d.proc'; 
opti donn 'pxrota3d.proc'; 
opti echo 1; 
opti dime 3 elem cub8; 
opti titr 'bott02'; 
opti trac psc ftra 'bott02_mesh.ps'; 
opti sauv form 'bott02.msh'; 
oeily = 0 -100000 0; 
* 
l = 0.6; 
ri = 0.1; 
h = 0.005; 
re = ri + h; 
* 
p0 = 0 0 0; 
p1 = 0 0 l; 
p2 = 0 0 (0 - ri); 
p3 = ri 0 0; 
p4 = ri 0 l; 
p8 = 0 0 (0 - ri - h); 
p9 = (ri + h) 0 0; 
p10 = (ri + h) 0 l; 
p1y = p1 plus (0 -1 0); 
p11 = p10 TOUR 70.0 p1 p1y; 
p14 = p4 TOUR 70.0 p1 p1y; 
x11 y11 z11 = coor p11; 
x14 y14 z14 = coor p14; 
p12 = x11 0 0.75; 
p13 = 0 0 0.75; 
p5 = p4 tour 80.0 p1 p1y; 
x5 y5 z5 = coor p5; 
p15 = x5 0 (z5 + h); 
p16 = x5 0 0.75; 
p6 = x5 0 0.725; 
p7 = 0 0 0.725; 
p17 = ri 0 (l*0.5); 
tol = 1.E-4; 
* 
c1 = p2 d 1 p8; 
p0y = p0 plus (0 -1 0); 
bot1 = c1 ROTA 18 90.0 p0 p0y; 
elim tol (bot1 et p3 et p9); 
* 
c2 = p3 d 1 p9; 
bot2 = c2 TRAN 60 (0 0 l); 
elim tol (bot2 et p10 et p4 et p17); 
* 
c3 = p4 d 1 p10; 
bot3 = c3 ROTA 15 70.0 p1 p1y; 
elim tol (bot3 et p11 et p14); 
* 
c4 = p5 c 3 p1 p14; 
c5 = p14 d 1 p11; 
c6 = p11 c 3 p1 p15; 
c7 = p15 d 1 p5; 
bot4 = DALL c4 c5 c6 c7 PLAN; 
* 
ca = p15 c 3 p1 p11; 
cb = p11 d 10 p12; 
cc = p12 d 3 p16; 
cd = p16 d 5 p6 d 5 p15; 
bot5 = DALL ca cb cc cd PLAN; 
* 
ce = p7 d 3 p6; 
cf = p6 d 5 p16; 

cg = p16 d 3 p13; 
ch = p13 d 5 p7; 
bot6 = DALL ce cf cg ch PLAN; 
* 
bott = bot1 et bot2 et bot3 et bot4 et bot5 et bot6; 
elim tol bott; 
* 
bott1a ier s1n = pxvolu3d (bot1 et bot2) 9 -90.0 c1 tol; 
list ier; 
bott1b ier s1n = pxvolu3d (bot3 et bot4 et bot5 et bot6) 9 -90.0 ch tol; 
list ier; 
bott1 = bott1a et bott1b; 
elim tol bott1; 
* 
bott2 = bott1 tour 90.0 p0 p1; 
bott3 = bott2 tour 90.0 p0 p1; 
bott4 = bott3 tour 90.0 p0 p1; 
bottle = bott1 et bott2 et bott3 et bott4; 
elim tol bottle; 
* 
trac cach bottle; 
* 
trac oeily bott; 
trac oeily qual bott; 
trac qual oeily (bot3 et bot4 et bot5 et bot6); 
trac oeily (bot3 et bot4 et bot5 et bot6); 
* 
bottl8 = bottle elem cub8; 
bottl6 = bottle elem pri6; 
list (nbno bottle); 
list (nbel bottle); 
list (nbel bottl8); 
list (nbel bottl6); 
* 
* internal surface (to apply the pressure) 
* 
icon1 = p2 c 18 p0 p3 d 60 p4; 
*icon2 = p4 c 15 p1 p14 c 3 p1 p5 d 1 p15 d 5 p6 d 3 p7; 
icon2 = p7 d 3 p6 d 5 p15 d 1 p5 c 3 p1 p14 c 15 p1 p4; 
trac (icon1 et icon2); 
isur1a ier = pxrota3d icon1 9 -90.0 c1 tol;  
list ier; 
ax = p7 d 1 p13; 
isur1b ier = pxrota3d icon2 9 90.0 ax tol;  
list ier; 
isur1 = isur1a et isur1b; 
isur2 = isur1 tour 90.0 p0 p1; 
isur3 = isur2 tour 90.0 p0 p1; 
isur4 = isur3 tour 90.0 p0 p1; 
isurf = isur1 et isur2 et isur3 et isur4; 
elim tol (isurf et bottle); 
trac cach isurf; 
* 
isurf4 = isurf elem qua4; 
isurf3 = isurf elem tri3; 
list (nbno isurf); 
list (nbel isurf); 
list (nbel isurf4); 
list (nbel isurf3); 
e17 = (bott1 elem appu larg p17) elem 1; 
list e17; 
* 
mesh = bottle et isurf; 
tass mesh; 
sauv form mesh; 
* 
fin; 

 

The EUROPLEXUS input file reads: 
 
bott02 
ECHO 
!CONV win 
TRID LAGR FAIL 0.0 
CAST FORM mesh 
DIME 
  PT3L 13652 CUB8 4860 PR6 216 CL3D 3708 CL3T 72 DEBR 4860 ZONE 5 
TERM 
GEOM  

  CUB8 bottl8  
  PR6  bottl6 
  CL3D isurf4 
  CL3T isurf3 
TERM 
COMP COUL rose LECT bottle TERM 
          vert LECT isurf  TERM 
     DEBR ROF 1.0 
          FILL PLEV 0 

               RO 8000. DRAG 1.0 OBJE LECT bottl8 TERM 
MATE VM23 RO 8000 YOUN 2.E11 NU 0.3 ELAS 2.E8 
          FAIL PEPS LIMI 0.05 
          TRAC 2 2.E8 0.001 2.02E8 1.001 
          LECT bottle TERM 
     IMPE PIMP RO 1.0 PRES 1.0 PREF 0.0 FONC 1 
          LECT isurf TERM 
FONC NUM 1 TABL 2 0.0 2.E7 1.0 2.E10  
ECRI DEPL VITE TFRE 1.E-4 
     FICH SPLI ALIC TFRE 1.E-6 
     FICH ALIC TEMP TFRE 1.E-6 
          POIN LECT p2 p3 p17 p4 p7 TERM 
          ELEM LECT e17 TERM 
OPTI NOTE LOG 1  
CALC TINI 0. TEND 0.35E-3 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 

ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR 1 'dy_p2' DEPL COMP 2 POIN LECT p2 TERM 
COUR 2 'dx_p3' DEPL COMP 1 POIN LECT p3 TERM 
COUR 3 'dx_p17' DEPL COMP 1 POIN LECT p17 TERM 
COUR 4 'dx_p4' DEPL COMP 1 POIN LECT p4 TERM 
COUR 5 'dy_p7' DEPL COMP 2 POIN LECT p7 TERM 
COUR 6 'vm_e17' ECRO COMP 2 ELEM LECT e17 TERM 
TRAC 1 2 3 4 5 AXES 1.0 'Displ. [M]' 
TRAC 6 AXES 1.0 'Von Mises [Pa]' 
FIN 

 
The fragmentation process and the velocities are shown in the next figures: 

 
 

BOTT05 
 
This is the second silulation, including both the structure and the fluid. The Cast3m 
mesh generation file reads: 
 
opti echo 0; 
opti donn 'pxordpoi.proc'; 
opti donn 'pxvolu3d.proc'; 
opti donn 'pxrota3d.proc'; 
opti echo 1; 
opti dime 3 elem cub8; 
opti titr 'bott05'; 
opti trac psc ftra 'bott05_mesh.ps'; 
opti sauv form 'bott05.msh'; 
oeil1 = 100000 100000 100000; 
oeily = 0 -100000 0; 
* 
l = 0.6; 
ri = 0.1; 
h = 0.005; 
re = ri + h; 
* 
p0 = 0 0 0; 
p1 = 0 0 l; 
p2 = 0 0 (0 - ri); 
p3 = ri 0 0; 
p4 = ri 0 l; 
p8 = 0 0 (0 - ri - h); 
p9 = (ri + h) 0 0; 
p10 = (ri + h) 0 l; 
p1y = p1 plus (0 -1 0); 
p11 = p10 TOUR 70.0 p1 p1y; 
p14 = p4 TOUR 70.0 p1 p1y; 
x11 y11 z11 = coor p11; 
x14 y14 z14 = coor p14; 
p12 = x11 0 0.75; 
p13 = 0 0 0.75; 
p5 = p4 tour 80.0 p1 p1y; 
x5 y5 z5 = coor p5; 
p15 = x5 0 (z5 + h); 
p16 = x5 0 0.75; 
p6 = x5 0 0.725; 
p7 = 0 0 0.725; 
p17 = ri 0 (l*0.5); 
tol = 1.E-4; 
* 
c1 = p2 d 1 p8; 
p0y = p0 plus (0 -1 0); 
bot1 = c1 ROTA 18 90.0 p0 p0y; 
elim tol (bot1 et p3 et p9); 
* 
c2 = p3 d 1 p9; 
bot2 = c2 TRAN 60 (0 0 l); 
elim tol (bot2 et p10 et p4 et p17); 
* 
c3 = p4 d 1 p10; 
bot3 = c3 ROTA 15 70.0 p1 p1y; 
elim tol (bot3 et p11 et p14); 
* 
c4 = p5 c 3 p1 p14; 
c5 = p14 d 1 p11; 
c6 = p11 c 3 p1 p15; 
c7 = p15 d 1 p5; 
bot4 = DALL c4 c5 c6 c7 PLAN; 
* 
ca = p15 c 3 p1 p11; 
cb = p11 d 10 p12; 
cc = p12 d 3 p16; 
cd = p16 d 5 p6 d 5 p15; 
bot5 = DALL ca cb cc cd PLAN; 
* 

ce = p7 d 3 p6; 
cf = p6 d 5 p16; 
cg = p16 d 3 p13; 
ch = p13 d 5 p7; 
bot6 = DALL ce cf cg ch PLAN; 
* 
bott = bot1 et bot2 et bot3 et bot4 et bot5 et bot6; 
elim tol bott; 
* 
bott1a ier s1n = pxvolu3d (bot1 et bot2) 9 -90.0 c1 tol; 
list ier; 
bott1b ier s1n = pxvolu3d (bot3 et bot4 et bot5 et bot6) 9 -90.0 ch tol; 
list ier; 
bott1 = bott1a et bott1b; 
elim tol bott1; 
* 
bott2 = bott1 tour 90.0 p0 p1; 
bott3 = bott2 tour 90.0 p0 p1; 
bott4 = bott3 tour 90.0 p0 p1; 
bottle = bott1 et bott2 et bott3 et bott4; 
elim tol bottle; 
* 
trac cach bottle; 
* 
trac oeily bott; 
trac oeily qual bott; 
trac qual oeily (bot3 et bot4 et bot5 et bot6); 
trac oeily (bot3 et bot4 et bot5 et bot6); 
* 
bottl8 = bottle elem cub8; 
bottl6 = bottle elem pri6; 
list (nbno bottle); 
list (nbel bottle); 
list (nbel bottl8); 
list (nbel bottl6); 
* 
* internal volume (to apply the pressure) 
* 
p2p = p2 plus p0; 
p3p = p3 plus p0; 
p4p = p4 plus p0; 
p14p = p14 plus p0; 
p5p = p5 plus p0; 
p15p = p15 plus p0; 
p6p = p6 plus p0; 
p7p = p7 plus p0; 
icon = p2p c 18 p0 p3p d 60 p4p c 15 p1 p14p c 3 p1 p5p d 1 p15p 
       d 5 p6p d 3 p7p d 20 p1 d 60 p0 d 20 p2p; 
isur = SURF icon plan; 
trac isur; 
* 
p18 = 0 0 -0.2; 
p19 = 0.2 0 -0.2; 
p20 = 0.2 0  0.9; 
p21 = 0   0  0.9; 
p16p = p16 plus p0; 
p13p = p13 plus p0; 
p12p = p12 plus p0; 
p11p = p11 plus p0; 
p10p = p10 plus p0; 
p9p = p9 plus p0; 
p8p = p8 plus p0; 
p22 = 0.2 0 0; 
econ1 = p9p c 18 p0 p8p d 10 p18 d 20 p19 d 20 p22 d 10 p9p; 
econ2 = p13p d 3 p16p d 3 p12p d 10 p11p c 15 p1 p10p d 60 p9p 
        d 10 p22 d 80 p20 d 20 p21 d 14 p13p; 

esur1 = SURF econ1 plan; 
esur2 = SURF econ2 plan; 
esur = esur1 et esur2; 
elim tol esur; 
trac esur; 
trac (isur et esur); 
*bsur = bott plus p0; 
bsur1 = bot1 plus p0; 
bsur2 = (bot2 et bot3 et bot4 et bot5 et bot6) plus p0; 
bsur = bsur1 et bsur2; 
elim tol (isur et bsur et esur); 
trac bsur; 
trac (isur et esur et bsur); 
* 
axis = p7p d 20 p1 d 60 p0 d 20 p2p; 
elim tol (axis et isur); 
ivol1 ier s1n = pxvolu3d isur 9 -90.0 axis tol; 
list ier; 
trac oeil1 cach ivol1; 
ivol2 = ivol1 tour 90.0 p0 p1; 
ivol3 = ivol2 tour 90.0 p0 p1; 
ivol4 = ivol3 tour 90.0 p0 p1; 
ivol = ivol1 et ivol2 et ivol3 et ivol4; 
elim tol ivol; 
trac oeil1 cach ivol; 
* 
besur1 = bsur1 et esur1; 
axis1 = p2p d 1 p8p d 10 p18; 
elim tol (axis1 et besur1); 
besur2 = bsur2 et esur2; 
axis2 = p21 d 14 p13p d 5 p7p; 
elim tol (axis2 et besur2); 
besur = besur1 et besur2; 
* 
bevol1a ier s1n = pxvolu3d besur1 9 -90.0 axis1 tol; 
list ier; 

bevol2a ier s1n = pxvolu3d besur2 9 -90.0 axis2 tol; 
list ier; 
bevol1 = bevol1a et bevol2a; 
elim tol bevol1; 
trac oeil1 cach bevol1; 
bevol2 = bevol1 tour 90.0 p0 p1; 
bevol3 = bevol2 tour 90.0 p0 p1; 
bevol4 = bevol3 tour 90.0 p0 p1; 
bevol = bevol1 et bevol2 et bevol3 et bevol4; 
elim tol bevol; 
trac oeil1 cach bevol; 
* 
flui = ivol et bevol; 
elim tol flui; 
* 
flui8 = flui elem CUB8; 
flui6 = flui elem PRI6; 
* 
eabs = enve flui; 
eabs4 = eabs elem QUA4; 
eabs3 = eabs elem TRI3; 
* 
list (nbno flui); 
list (nbel flui8); 
list (nbel flui6); 
list (nbel eabs4); 
list (nbel eabs3); 
* 
e17 = (bott1 elem appu larg p17) elem 1; 
list e17; 
* 
mesh = bottle et flui et eabs; 
tass mesh; 
sauv form mesh; 
* 
fin; 

 

The EUROPLEXUS input file reads: 
 
bott05 
ECHO 
!CONV win 
TRID ALE FAIL 0.0 
CAST FORM mesh 
DIME 
  PT3L 112531 
  CUB8   4860 PR6   216 
  FL38  92988 FL36 6480 
  CL3Q   4968 
  CL3I     72 
  DEBR   4860 
  ZONE 7 
  NALE 1 NBLE 1 
TERM 
GEOM  
  CUB8 bottl8  
  PR6  bottl6 
  FL38 flui8 
  FL36 flui6 
  CL3Q eabs4 
  CL3I eabs3 
TERM 
COMP DEBR ROF 1.0 
          FLUI LECT bevol TERM DGRI DELE 1.5 
          FILL PLEV 0 
               RO 8000. DRAG 1.0 OBJE LECT bottl8 TERM 
     GROU 1 'part' LECT 109585 PAS 1 114444 TERM 
     COUL rose LECT bottle TERM 
          roug LECT ivol   TERM 
          turq LECT bevol  TERM 
          jaun LECT eabs   TERM 
GRIL LAGR LECT bottle part TERM 
MATE VM23 RO 8000 YOUN 2.E11 NU 0.3 ELAS 2.E8 
          FAIL PEPS LIMI 0.05 
          TRAC 2 2.E8 0.001 2.02E8 1.001 
          LECT bottle TERM 
     FLUT RO 1.0 EINT 2.5E5 GAMM 1.4 PB 0 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 PREF 1.E5 

          LECT bevol TERM 
     FLUT RO 100.0 EINT 5.0E5 GAMM 1.4 PB 0 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 PREF 1.E5 
          LECT ivol TERM 
     IMPE ABSI LECT eabs TERM 
LINK COUP FLSR STRU LECT bottle TERM 
               FLUI LECT flui   TERM 
               PHIS 0.3D0 
               PHIF 0.2D0 
               HGRI 0.01D0 
ECRI DEPL VITE TFRE 1.E-4 
     FICH SPLI ALIC TFRE 1.E-6 
     FICH ALIC TEMP TFRE 1.E-6 
          POIN LECT p2 p3 p17 p4 p7 TERM 
          ELEM LECT e17 TERM 
OPTI NOTE LOG 1  
CALC TINI 0. TEND 0.35E-3 
fin 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR 1 'dy_p2' DEPL COMP 2 POIN LECT p2 TERM 
COUR 2 'dx_p3' DEPL COMP 1 POIN LECT p3 TERM 
COUR 3 'dx_p17' DEPL COMP 1 POIN LECT p17 TERM 
COUR 4 'dx_p4' DEPL COMP 1 POIN LECT p4 TERM 
COUR 5 'dy_p7' DEPL COMP 2 POIN LECT p7 TERM 
COUR 6 'vm_e17' ECRO COMP 2 ELEM LECT e17 TERM 
TRAC 1 2 3 4 5 AXES 1.0 'Displ. [M]' 
TRAC 6 AXES 1.0 'Von Mises [Pa]' 
FIN 

 
 
The fragmentation process and the velocities are shown in the next figures: 
 
 
 
 
 

 


